# $Id: PKGBUILD 71600 2012-05-30 16:08:38Z bpiotrowski $
# Maintainer:  Bartłomiej Piotrowski <nospam@bpiotrowski.pl>
# Contributor: Angel Velasquez <angvp@archlinux.org> 
# Contributor: Geoffroy Carrier <geoffroy.carrier@koon.fr>

pkgname=lxterminal
pkgver=0.1.11
pkgrel=2
pkgdesc="VTE-based terminal emulator (part of LXDE)"
arch=('i686' 'x86_64')
license=('GPL2')
url="http://lxde.org/"
groups=('lxde')
depends=('fontconfig' 'libx11' 'glib2' 'vte' 'gtk2')
makedepends=('pkgconfig' 'intltool')
source=(http://downloads.sourceforge.net/sourceforge/lxde/${pkgname}-${pkgver}.tar.gz
        shift-keybinds.patch)
md5sums=('fd9140b45c0f28d021253c4aeb8c4aea'
         '7eff715e02181b006b1912cd041b7f1a')

build() {
	cd "$srcdir"/$pkgname-$pkgver

	patch -Np1 -i "$srcdir"/shift-keybinds.patch

	./configure --sysconfdir=/etc --prefix=/usr
	make
}

package() {
	cd "$srcdir"/$pkgname-$pkgver
	make DESTDIR="$pkgdir" install
}

